

## 補遺 3

パラメータ  $k_1, k_2, p_1, p_2$  と有効雨量系列既知の  
流出計算プログラムと計算例

### 補遺3 パラメータ $k_1, k_2, p_1, p_2$ と有効雨量系列既知の流出計算プログラムと計算例

当モデルは非線形であり、理論解が得られないため数値解法によらざるを得ないことは前述した通りである。ここではモデルパラメータ  $p_1, p_2, k_1, k_2$  が既知であるときの計算の流れを説明する。また、計算フロー、プログラム (BASIC)、テストデータによる出力例を載せておくので、本文中の理論が実用上どのようにプログラムに反映されているか流れを追ってみることをすすめる。

$$s = k_1 q^{p_1} + k_2 \frac{d}{dt}(q^{p_2}) \dots\dots\dots (1)$$

$$\frac{ds}{dt} = r - q \dots\dots\dots (2)$$

(1), (2)式で示された非線形方程式を線形方程式に変換し、差分表示すると次のようになる。

$$X_{k+1} = \Phi_k X_k + \Gamma_k B_k \dots\dots\dots (3)$$

$$\begin{bmatrix} x_1 \\ x_2 \end{bmatrix}_{k+1} = \begin{bmatrix} \phi_1 & \phi_2 \\ \phi_3 & \phi_4 \end{bmatrix} \begin{bmatrix} x_1 \\ x_2 \end{bmatrix}_k + \begin{bmatrix} \gamma_1 \\ \gamma_2 \end{bmatrix} (b_2)_k \dots\dots\dots (4)$$

(4)式中の要素  $\phi_1, \phi_2, \phi_3, \phi_4, \gamma_1, \gamma_2, b_2$  は本文中 P 8 ~ P 9 に示されている。これらの要素が求められれば、 $x_1, x_2$  は次のように求められる。

$$\begin{cases} (x_1)_{k+1} = \phi_1(x_1)_k + \phi_2(x_2)_k + \gamma_1(b_2)_k \\ (x_2)_{k+1} = \phi_3(x_1)_k + \phi_4(x_2)_k + \gamma_2(b_2)_k \end{cases} \dots\dots\dots (5)$$

ここで、 $x_1 = q^{p_1}, x_2 = \frac{d}{dt}(q^{p_2})$

本文中 P 8 ~ 9 に示されるように、 $\phi, \gamma$  の係数は級数に展開されている。一般にある関数を級数展開したとき、項数を多くすれば解の精度は向上する。しかし、計算時間間隔  $T$  を 1 時間とするよりも、1 時間の 5 分の 1 あるいは 10 分の 1 に設定すれば、 $T$  のべき乗に関する項は項数が多くなるにつれ無視できる程度に小さくなることが予想される。また、そうすることにより、計算しなければならない項が少なくて済む利点がある。ここでは、1 時間を 5 分割し、それを計算ステップとして逐次計算を行う。

計算時間間隔と計算値の出力に関しては図-3・1を参照されたい。計算時間間隔  $T$  ごとに(5)式を計算し、 $x_1, x_2$  を逐次求めていく。流出高の出力は図-3・1の黒丸の時点 (サンプリング時間間隔ごと) で  $q = x_1^{1/p_1}$  を計算して出力してやればよい。

ここではパラメータ  $p_1, p_2, k_1, k_2$  が既知であるとき、有効降雨を与えたときの流量 (直接流出高) をどのように求めていくかをフロー図として示し、その後プログラム、テストデータとその出力例を示す。






図-3・1 計算値の出力

$p_1, p_2, k_1, k_2$ が既知のときの計算フロー



サブルーチン "GESTO" のフロー



パラメータが既知のときの流量計算プログラム

```

30 DIM R(50),Q(50)
40 P1=.6:P2=.4648
50 K1= 6.3459 :K2=10.552
60 NR=15:NQ=31
70 ' R DATA
80 FOR I=1 TO NR:READ R(I):NEXT I
90 DATA 0.5940,0.7920,0.6930,0.3960,0.4950,0.9900,0.6930,0.3960
95 DATA 1.1880,1.9799,3.0689,4.0588,3.0689,1.0890,0.6930
100 '
110 T=1/5 : ' CULCULATION STEP 1/5(HOUR)
120 T2=T*T:T3=T2*T:T4=T3*T
130 NL=INT(1/T+1E-03)
140 '
150 P3=P1/P2:K3=K1/K2
160 PRINT
170 LPRINT USING "NL=##";NL
180 LPRINT USING "NR=##  NQ=##";NR;NQ
190 LPRINT USING "K1=###.###  K2=###.###";K1;K2
200 LPRINT
210 LPRINT "NO      YR      CQ"
220 LPRINT "      (MM/H)  (MM/H)"
230 FOR J=1 TO NQ
240 Y1=X(1) :Y2=X(2)
250 FOR I=1 TO NL
260 GOSUB 1000:'SUBROUTINE GESTO
470 NEXT I
480 Q(J)=X(1)^(1/P2)
490 LPRINT USING "##  ##.###  ##.###";J;R(J);Q(J)
500 NEXT J
510 END
1000 'SUBROUTINE GESTO
1010 IF Y1=0 THEN GOTO 1060
1020 A=Y1^(P3-2)
1030 C=Y1^(1/P2-1)
1040 D=Y1^(P3-1)
1050 E=Y1^(1/P2)
1060 A1=-K3*P3*(P3-1)*A*Y2-1/K2/P2*C
1070 A2=-K3*P3*D
1080 A3=A1+A2*A2:A4=A1+A3
1090 B2=K3*P3*(P3-1)*D*Y2+1/K2*(1/P2-1)*E+R(J)/K2
1100 ' FAI MATRIX
1110 F1=1+.5*A1*T2+1/6*A1*A2*T3+1/24*A1*A3*T4
1120 F2=T*(1+.5*A2*T+1/6*A3*T2+1/24*A2*A4*T3)
1130 F3=A1*F2
1140 F4=1+A2*T+.5*A3*T2+1/6*A2*A4*T3+1/24*(A1*A3+A2*A2*A4)*T4
1150 ' GAMMA MATRIX
1160 G2=T2*(.5+1/6*A2*T+1/24*A3*T2)
1170 G4=F2
1180 '
1190 Y(1)=F1*Y1+F2*Y2+G2*B2
1200 Y(2)=F3*Y1+F4*Y2+G4*B2
1210 Y1=Y(1):Y2=Y(2)
1220 FOR L=1 TO 2:X(L)=Y(L):NEXT L
1230 RETURN

```

テストデータ及び出力例

NL= 5  
 NR=15 NQ=31  
 K1= 6.3459 K2= 10.5520

NO	YR (MM/H)	CQ (MM/H)
1	0.594	0.000
2	0.792	0.008
3	0.693	0.035
4	0.396	0.083
5	0.495	0.139
6	0.990	0.217
7	0.693	0.313
8	0.396	0.388
9	1.188	0.473
10	1.980	0.642
11	3.069	0.988
12	4.059	1.608
13	3.069	2.309
14	1.089	2.561
15	0.693	2.305
16	0.000	1.814
17	0.000	1.319
18	0.000	0.940
19	0.000	0.675
20	0.000	0.496
21	0.000	0.375
22	0.000	0.291
23	0.000	0.232
24	0.000	0.189
25	0.000	0.157
26	0.000	0.132
27	0.000	0.113
28	0.000	0.098
29	0.000	0.086
30	0.000	0.076
31	0.000	0.068

NL : 1時間の分割数  
 NR : 雨量データ個数  
 NQ : 計算流量個数  
 K1, K2 : モデルパラメータ  $k_1, k_2$   
 YR : 有効雨量 (mm/hr)  
 CQ : 直接流出高 (mm/hr)




図-3・2